
//***************************************************************
//* MACRO NAME: LINENUM.WCM
//* PURPOSE: To goto a specific line and character in a macro.
//***************************************************************
Application (A1; "WordPerfect"; Default; "UK")

TheLine:=1
TheChar:=1

DialogDefine("LINE"; 50; 50; 159; 83; 1+2+16; "Macro Line Finder")
DialogAddText("LINE"; "T1"; 8; 10; 92; 13; 0; "Enter the &line number:")
DialogAddCounter("LINE"; "C1"; 104; 8; 40; 14; 0; TheLine; 1; 99999; 1)
DialogAddText("LINE"; "T2"; 8; 27; 92; 20; 0; "Enter the &character number:")
DialogAddCounter("LINE"; "C2"; 104; 25; 40; 15; 0; TheChar; 1; 99999; 1)
Display(On!)
DialogDisplay("LINE"; "C1")
Display(Off!)
Result:=MacroDialogResult
DialogDestroy("LINE")
If(Result<>1)

Go(End@)
EndIf
WaitMessage:="Searching for line and character"
Call(WaitMessage@)
OnError(Quit@)
OnCancel(Quit@)
DormHRt:=0
PosDocTop()
SearchFindWholeWordsOnly(No!)
MatchPositionAfter()
SearchString("[HRt]")
OnNotFound(Quit@)

Label(Search@)
For(i;1;i<TheLine;i+1)

SearchNext(Regular!)
If(?LeftCode=135)

DormHRt:=DormHRt+1
EndIf

EndFor

If(DormHRt>0)
TheLine:=DormHRt+1
DormHRt:=0
Go(Search@)

EndIf



For(i; 1; i<TheChar; i+1)
If(?RightCode=204)

Go(Quit@)
EndIf
PosCharNext()

EndFor

Label(Quit@)
Call(KillWaitMessage@)
Label(End@)
Quit

//***************************************************************
//* ROUTINE:  WaitMessage@
//* INPUT VARIABLES:  WaitMessage
//* OUTPUT VARIABLES:  WaitMessageResult
//* DESCRIPTION:  Presents a Please Wait message to user.  Call KillWaitMessage to turn 
it off.
//***************************************************************
Label(WaitMessage@)
WaitTitle:="WAITPROMPT"
WaitMessageWidth:=105
DialogDefine(WaitTitle; 50; 50; WaitMessageWidth+22; 43; 16+256+8+2; WaitTitle)
RegionShowWindow(WaitTitle+".CancelBttn"; Hide!)
DialogAddText(WaitTitle; "T1"; 8; 8; 46; 10; 1; "Please Wait...")
DialogAddText(WaitTitle; "T2"; 8; 19; WaitMessageWidth; 10; 1; WaitMessage)
DialogDisplay(WaitTitle; 1; WaitDlgCallBack@)
Return

Label(WaitDlgCallBack@)
If(WaitDlgCallBack[3] = "CancelBttn")

Assert(CancelCondition!)
EndIf
Return
//***************************************************************

//***************************************************************
//* ROUTINE:  KillWaitMessage@
//* INPUT VARIABLES:  none
//* OUTPUT VARIABLES:  none
//* DESCRIPTION:  Turns off wait message.
//***************************************************************
Label(KillWaitMessage@)
DialogDestroy(WaitTitle)
Return



//***************************************************************


